Calcul des adjoints et programmation paresseuse

Jerzy Karczmarczuk

Dept. d'Informatique, Université de Caen, France
(`mailto:karczma@info.unicaen.fr`*)*

Résumé

Nous présentons une réalisation purement fonctionnelle et paresseuse de la technique du mode inverse de différentiation algorithmique. Cette technique, qui permet de calculer de manière précise et efficace les dérivées des expressions numériques dans un programme, est devenue indispensable dans plusieurs branches de programmation scientifique. Le mode inverse ou adjoint demande souvent l'usage des lourds paquetages extérieurs, et du pré-traitement du programme source. Grâce aux techniques de programmation paresseuse nous montrons comment intégrer de manière facile et transparente la construction des dérivées adjointes dans le programme même. Le résultat est pratiquement utilisable même si plusieurs optimisations seront nécessaires pour rendre le paquetage utilisable dans des cas plus sérieux.

(Version préliminaire)

1. Introduction

1.1. Différentiation algorithmique

Dans ce travail nous élaborons une technique particulière de différentiation algorithmique des programmes numériques. Notre implantation est purement fonctionnelle, et exploite la sémantique paresseuse de manière assez agressive. Le paquetage a été réalisé en Haskell et testé avec l'interprète Hugs. Nous attendons que le lecteur soit famillier avec le concept de la programmation paresseuse, et qu'il puisse lire des programmes en Haskell.

En général, les outils de différentiation algorithmique (DA), ou «automatique» servent à calculer dans un programme *numérique* les dérivées, gradients, Jacobiens, etc. des expressions par rapport à un ou plusieurs objets considérés comme des *variables*, ce qui permet par exemple de calculer la dérivée d'une procédure par rapport à son paramètre pour une valeur numérique concrète. Le mot «numérique» signifie ici deux choses :

- aucune manipulation symbolique des expressions n'a lieu, les variables n'ont pas de noms comme dans un programme interprété par un système de calcul formel, et les expressions (p. ex. `x*y`) ne sont pas des arbres qui puissent être destructurés en composantes, ce qui permettrait leur traitement symbolique ;
- on obtient la valeur numérique (réelle, complexe, etc.) de l'expression dérivée, calculée en fonction d'autres valeurs numériques définissant son contexte.

On n'utilise pas des différences finies. Les techniques de DA sont *exactes*, c'est-à-dire, les dérivées sont calculées avec la même précision que toutes les autres expressions numériques, ce qui est determiné par les propriétés du processeur et les librairies de fonctions sur les nombres flottants.

La DA est un vénérable et important domaine de recherche et d'applications dans le monde d'ingénierie et du calcul scientifique, voir p. ex. [1, 2, 3, 4, 5]. Elle est utilisée pour analyser la stabilité des équations différentielles, pour résoudre quelques problèmes variationnels et dans des centaines d'autres cas. Même dans les mathématiques discrètes la différentiation peut servir comme un outil permettant de calculer les coefficients combinatoires à partir de leur fonction génératrice [6].

La technique est basée sur l'observation suivante : le calcul différentiel est algorithmiquement si simple, que toute manipulation des expressions numériques permettant de calculer les dérivées, peut être *facilement* effectuée par le compilateur (ou un logiciel de pré-traitement de la source) capable d'étendre la sémantique du programme original, en accord avec les règles du calcul différentiel. Ainsi, avec chacune expression originale on peut calculer simultanément sa dérivée à partir des constantes dont les dérivées sont égales à zéro, et la (ou les) *variables*, dont les dérivées sont triviales aussi. Le programme étendu suit la règle de Leibniz : $(ef)' = e'f + ef'$, et, plus généralement – la règle d'enchaînement : $(f(g(x))' = f'(g(x)) \cdot g'(x)$, et en composant des expressions plus complexes, calcule leurs dérivées pour les mêmes valeurs numériques de toutes les variables du programme. L'implantation la plus primitive consiste à surcharger l'arithmétique sur des paires de valeurs numériques : la paire (e, e') est une valeur étendue représentant une expression et sa dérivée (pour simplicité nous discutons ici le cas 1-dimensionnel, facile à généraliser). Dans ce nouveau domaine toutes les constantes explicites c prendront la forme $(c, 0)$, et la *variable* distinguée x deviendra $(x, 1)$. Les opérations arithmétiques seront surchargées par un «lifting» : $(e, e') + (f, f') = (e + f, e' + f')$; $(e, e') \cdot (f, f') = (e \cdot f, e' \cdot f + e \cdot f')$; $\exp(e, e') = \big(\exp(e), \exp(e) \cdot e'\big)$, etc. La dérivée est tout simplement le second élément d'une telle paire. Le seul problème ici est le fait qu'une telle extension ne définit pas une algèbre close, on ne peut pas facilement calculer la seconde dérivée. Bien sûr, de telles réalisations qui demandent uniquement la possibilité de surcharger les opérateurs arithmétiques, existent depuis longtemps.

Dans [7] nous avons étendu l'arithmétique sur des séquences infinies $[e, e', e'', e^{(3)}, \ldots]$ représentant les expressions avec *toutes* leur dérivées, ce qui a permis de définir une algèbre différentielle close, c'est-à-dire un domaine qui dispose de la panoplie arithmétique standard, et d'un opérateur de *dérivation* $\mathbf{d}$, linéaire, et satisfaisant la règle de Leibniz : $\mathbf{d}(ef) = e\mathbf{d}f + (\mathbf{d}e)f$. Dans ce domaine la «variable» x est une structure équivalente à la liste $[x, 1, 0, 0, \ldots]$, et les constantes c deviennent évidemment des listes $[c, 0, 0, \ldots]$. Voici quelques définitions de l'arithmétique étendue dans ce domaine. Pour $e = (e_0 : \hat{e})$ et $f = (f_0 : \hat{f})$, où l'opérateur $(:)$ est un constructeur de listes (il forme e en ajoutant e_0 devant la liste $\hat{e}$)) nous aurons

$$\begin{aligned}
e + f &= (e_0 + f_0 : \hat{e} + \hat{f}) \\
e \cdot f &= (e_0 \cdot f_0 : e\hat{f} + \hat{e}f) \\
1/e &= w \quad \textbf{où} \quad w = (1/e_0 : -\hat{e} \cdot w^2) \\
\exp(e) &= w \quad \textbf{où} \quad w = (\exp(e_0) : w\hat{e}) \\
\sqrt{e} &= w \quad \textbf{où} \quad w = (\sqrt{e_0} : 1/2 \cdot \hat{e}/w \\
\log(e) &= (\log(e_0) : \hat{e}/e)
\end{aligned} \tag{1}$$

etc. L'opérateur de dérivation $\mathbf{d} : e \to \hat{e}$ récupère la queue de la liste, et ainsi nous aurons «gratuitement» aussi les dérivées de degrée supérieur. Dans un langage paresseux $\mathbf{d}$ n'est pas trivial, car il peut forcer l'évaluation des expressions différées qui constituent cette queue. On note que les expressions ci-dessus sont co-récursives, «ouvertes», et qu'il faut éviter de demander la valeur des éléments dont on n'a pas besoin (par exemple d'afficher la liste complète), même si dans notre implantation on évite la construction des listes infinies triviales en utilisant une structure de données spéciale, où la liste infinie de zéros est remplacée par une constante spéciale.

La méthode présentée ici constitue le *mode direct* de la différentiation algorithmique. Le reste du travail décrit une technique alternative, dite «inverse», qui dans quelques contextes semble être plus naturelle, et parfois, surtout dans le cas multi-dimensionnel creux, aussi plus efficace. La généralisation de la méthode directe aux structures régulières : tenseurs et formes différentielles en N dimensions peut être trouvée dans l'article [8].

1.2. Mode inverse de différentiation

Les généralisations naturelles des listes infinies $e = (e_0 : \hat{e})$ pour plusieurs dimensions seraient des arbres $e = (e_0, [\hat{e}_1, \ldots, \hat{e}_n])$, avec $\hat{e}_k = (\partial e/\partial x_k, [\ldots \text{dérivées de } e'_k \ldots])$. La propagation de telles structures

pendant l'exécution du programme peut être assez coûteuse. Cependant, dans de très nombreuses applications, le nombre de résultats intéressants peut être très inférieur au nombre de variables indépendantes. Ceci est typique pour l'analyse de la sensibilité des processus techniques et naturels (dépendance de la solution finale de l'ensemble des paramètres du système et des conditions initiales). Exemples : météorologie et océanographie, diagnostic des réacteurs nucléaires, développement de la biosphère, etc. Dans de tels cas, les équations différentielles décrivent l'évolution du système dans un espace de plusieurs dimensions, mais cet espace n'a pas de propriétés «géométriques» régulières, les expressions intermédiaires dépendent d'habitude d'un nombre petit de paramètres (le problème est «creux»), et à la fin nous demandons un petit nombre de résultats, parfois un seul, par exemple la température du réacteur en fonction de ses nombreux paramètres d'exploitation.

Comme il est précisé dans de nombreux textes sur DA, dans ces circonstances il n'est pas nécessaire de maintenir toutes les dérivées partielles des expressions intermédiaires par rapport aux variables indépendantes. Il suffit de définir pour chaque variable (initiale ou intermédiaire) son **adjoint** $\overline{e}$ – la dérivée du *résultat final* par rapport à cette variable[1]. Il est évident qu'au moment de la définition d'une variable intermédiaire dans le programme, le résultat final n'est pas connu, et l'adjoint ne peut être effectivement calculé. Les paquetages de DA qui exploitent cette technique sont très compliqués., Nous allons montrer que la sémantique paresseuse simplifie l'implantation du mode inverse de manière spectaculaire, même si pour pouvoir l'appliquer aux problèmes réels, un sérieux travail d'optimisation semble encore nécessaire.

Décrivons la dérivation du mode inverse d'une manière plus formelle. Supposons que $(x_1, x_2, \ldots, x_M)$ constitue la collection des variables indépendantes dans le programme. Si à l'aide d'un «oracle» toutes les conditions dynamiques dans le programme sont résolues, les branchements effectués et les boucles aplâties, le programme peut être modélisé par l'enchaînement des définitions fonctionnelles :

$$\begin{aligned} x_{M+1} &\leftarrow f_{M+1}(x_1, \ldots, x_M)\,, \\ x_{M+2} &\leftarrow f_{M+2}(x_1, \ldots, x_{M+1})\,, \\ &\ldots \\ R = x_N &\leftarrow f_N(x_1, \ldots, x_{N-1})\,, \end{aligned} \tag{2}$$

où, pour l'homogénéité de la notation toutes les variables intermédiaires portent les noms «x_p». Naturellement, cet ensemble peut être complété par les affectations $x_k \leftarrow f_k()$, où f_k est une fonction constante qui fournit la valeur initiale de la variable en question pour $k \leq M$. Un petit nombre d'équations (2), peut-être seulement la dernière, qui définit R, spécifient les résultats du programme. Les fonctions f sont typiquement très «creuses», et se réduisent aux opérateurs unaires ou binaires.

Pour chaque affectation $g \leftarrow f(e_1, e_2, \ldots, e_k)$ les adjoints *des arguments à droite* sont obtenus par

$$\overline{e}_k \leftarrow \overline{e}_k + \overline{g}\frac{\partial f}{\partial e_k}\,. \tag{3}$$

Ceci n'est plus une définition (fonctionnelle) d'une variable intermédiaire, mais une structure impérative : la mise à jour d'une variable existante, puisque $\overline{e}_k$ peut figurer dans plusieurs endroits du programme-source. Pire, même si les instructions adjointes peuvent être facilement compilées dans le contexte de l'instruction originale, *elles ne peuvent y être exécutées*, car $\overline{g}$ sera connu plus tard, quand g sera *utilisé*.

Les paquetages qui exploitent le mode inverse, p. ex. [4, 9] et plusieurs autres, exécutent le programme étendu en deux étapes. D'abord le programme d'origine est exécuté, les valeurs de toutes les variables «normales» calculées, et en parallèle toutes les instructions adjointes sont stockées sur une structure de données linéaire, dite «bande» (normalement un fichier). Après avoir calculé le résultat final, la bande est lue et exécutée (interprétée) à l'envers, de la fin jusqu'au début, où se trouvent les premières affectations des adjoints des variables indépendantes.

Cette procédure est visiblement lourde, et un pré-traitement important du code-source est nécessaire, si le programme est écrit dans un langage classique impératif.

1. Ceci n'a aucun rapport avec les adjoints dans la théorie des catégories

Par exemple, si on veut calculer $z'(x)$, où z est défini par le programme :

$$y = \sin(x); \quad z = y^2 - x/y\,, \tag{4}$$

il faut d'abord calculer y et z pendant la phase «avant» du programme, ensuite initialiser les adjoints : $\overline{z} \leftarrow 1$; $\overline{x} \leftarrow 0$; $\overline{y} \leftarrow 0$, et renverser le flot de contrôle.

$$\begin{aligned} z = y^2 - x/y; \qquad \text{donne} \qquad & \overline{x} \leftarrow \overline{x} + \overline{z}(-1/y); \\ & \overline{y} \leftarrow \overline{y} + \overline{z}(2y + x/y^2); \\ y = \sin(x); \qquad \text{donne} \qquad & \overline{x} \leftarrow \overline{x} + \overline{y}\cos(x)\,. \end{aligned} \tag{5}$$

Finalement $\overline{x} = -1/\sin(x) + \cos(x)\left(2\sin(x) + x/\sin(x)^2\right)$ est la valeur de dz/dx.

Voici la dérivation générale de cet algorithme. Les dérivées sont définies par les enchaînements satisfaits par les matrices de Jacobi :

$$\mathbf{J}_{ik} \equiv \frac{dx_i}{dx_k} = \delta_{ik} + \sum_{j=k}^{i-1} \frac{\partial f_i}{\partial x_j}\frac{dx_j}{dx_k}\,. \tag{6}$$

La forme matricielle de cette équation est : $\mathbf{J} = \mathbf{I} + \mathbf{DJ}$, où

$$\mathbf{D}_{ik} \equiv \frac{\partial f_i}{\partial x_k} = \begin{pmatrix} 0 & 0 & 0 & \dots \\ \partial f_2/\partial x_1 & 0 & 0 & \dots \\ \partial f_3/\partial x_1 & \partial f_3/\partial x_2 & 0 & \dots \\ \vdots & \vdots & \vdots & \ddots \end{pmatrix}. \tag{7}$$

Si on commence par x_{M+1}, et si on suit la chaîne jusqu'à x_N, on calcule itérativement $\mathbf{J}_{ik}$, et ceci constitue la méthode directe. Mais on peut (en principe) calculer d'abord les dernières dérivées partielles, et propager les valeurs vers l'«arrière». Sur le plan d'efficacité ceci peut être intéressant, surtout si on a besoin seulement de la dernière ligne de la matrice de Jacobi, c'est-à-dire les éléments $\overline{x}_k = \mathbf{J}_{Nk} = dx_N/dx_k$. Il est évident que les matrices $\mathbf{J}$ and $\mathbf{D}$ commutent, et si l'équation définissant $\mathbf{J}$ est réécrite en sa forme adjointe : $\mathbf{J} = \mathbf{I} + \mathbf{JD}$, ou

$$\mathbf{J}_{ik} = \delta_{ik} + \sum_{j=k+1}^{i} \mathbf{J}_{ij}\mathbf{D}_{jk}\,, \tag{8}$$

la dernière ligne satisfait

$$\overline{x}_k = \delta_{Nk} + \sum_{j=k+1}^{N} \overline{x}_j \mathbf{D}_{jk}\,. \tag{9}$$

On voit immédiatemment le problème, l'adjoint $\overline{x}_k$ a besoin de $\overline{x}_l$ pour $l > k$. La machinerie qui calcule les adjoints n'est pas seulement lourde, elle est aussi dangereuse. Si le programme exécute une boucle, chaque réaffectation d'une variable intermédiaire (ou création d'une nouvelle instance de cette variable, si la boucle est réalisée par la récursivité terminale), engendre des *nouvelles* instructions adjointes, qui devront être mémorisées sur la «bande». Elle peut devenir très longue, et plusieurs optimisations seraient essentielles dans des cas plus complexes [10, 11], mais leur implantation automatique est difficile. Le reste de l'article est consacré à l'implantation fonctionnelle paresseuse du mode inverse. Nous voulons implanter ce mode de manière simple, transparente et pratique pour un utilisateur intéressé par le calcul scientifique, et pour l'instant nous ne discutons pas ces optimisations.

2. Application de la sémantique non-stricte

2.1. Transformateurs d'états et programmation paresseuse

Notre approche est basée sur les techniques monadiques de programmation fonctionnelle. Bien sûr, nous ne pouvons pas discuter ici les monades dans leur généralité, rappelons cependant l'idée essentielle de la

technique monadique qui permet d'implanter de manière fonctionnelle les effets de bord : la monade ST (*State Transformer*), décrite p. ex. dans [12].

Dans un programme fonctionnel l'évaluation d'une expression est «pure», on obtient une valeur appartenant, disons, à un type dénoté symboliquement par **a**, et c'est tout. Pour modéliser la notion d'état qui subit des modifications pendant l'exécution du programme, on remplace les expressions de type **a** par des objets fonctionnels de type **ST a** ≡ **s -> (a,s)**, où **s** dénote le type des objets représentant l'état : par exemple un compteur incrémenté à chaque évaluation d'une variable, une chaîne de caractères qui contient le trace de l'exécution du programme, etc. Si une expression e est neutre par rapport au changement de cet état, elle deviendra une fonction qui laisse l'état intact : **\s -> (e,s)**. (Rappelons qu'en Haskell le symbole «****» dénote λ.) Formellement aucune «transformation» n'a lieu : un opérateur agit sur l'état initial, et créé un autre état, appelé final. Les deux sont des données ordinaires. Toutefois, si le compilateur peut prouver qu'après la création de l'état final, le programme n'accède plus à l'original, il peut optimiser leur gestion en modifiant l'état initial sur place. Ceci est l'essentiel de l'approche monadique à la programmation impérative dans un langage fonctionnel.

Toute fonction qui agissait sur les expressions et produisait des valeurs du même type, doit maintenant engendrer les transformateurs d'état. Comment elle le fait, dépend de la fonction, nous pouvons cependant définir de manière universelle son «lifting», l'action d'une telle fonction sur un transformateur **m** de type **ST a**. Dans la définition ci-dessous l'opérateur **>=>** dénote l'application «liftée» d'une fonction **f** à un objet de ce type :

```
f >=> m =
 \s_init ->
     let (x,s_mid) = m s_init
         (y,s_final) = (f x) s_mid
     in  (y,s_final)
```

L'interprétation est simple : le résultat est une fonction qui doit agir sur un état initial. D'abord le transformateur **m** agit sur cet état, et engendre un état intermédiaire combiné avec **x** – l'argument effectif de la fonction **f**. Cette fonction agissant sur **x** crée un transformateur qui change l'état intermédiaire en final.

Dans [12] Philip Wadler a proposé une modification apparemment très bizarre de la composition des objets **ST** – les transformateurs dont l'enchaînement *propage l'état vers l'arrière dans le temps*. Voici la définition de l'application étendue :

```
f >=> m =
 \s_final ->
     let (x,s_init) = m s_interm
         (y,s_interm) = (f x) s_final
     in  (y,s_init)
```

Le résultat est une fonction qui agit sur l'état final. L'opérateur qui touche cet état est le transformateur créé par l'action de la fonction **f** sur son argument. Ce transformateur rend la valeur **y** qui conceptuellement représente $f(x)$, et accessoirement engendre l'état intermédiaire, qui sera consommé par **m** et transformé en l'état initial. Ceci n'est pas un simple changement de noms. Observons que l'argument **x** pour **f** est préparé par **m** agissant sur l'état intermédiaire créé par **f**. Les définitions ci-dessus sont circulaires, les données sont réciproquement dépendantes, et une telle intrication admet une solution (un programme qui fonctionne) seulement si la sémantique des appels fonctionnels est paresseuse, si – par exemple – **m** n'a pas immédiatemment besoin de **s_interm** pour récupérer et retourner **x**.

2.2. *Intermezzo :* propagation des attributs dans la compilation

A priori il est difficile de trouver des applications immédiates pour une telle «machine à voyager dans le temps» qui est loin d'être intuitive. Certes, les programmes circulaires qui exploitent la sémantique non-stricte pour optimiser la gestion de données par un programme circulaire sont connus depuis longtemps [13], mais ils ont

un goût plutôt artificiel. Dans un programme traditionnel les dépendences entre les données et le flot de contrôle sont synchrones. Il existe cependant un contexte, où les dépendances entre les entités sont parfois circulaires : la propagation des attributs sémantiques lors de l'analyse syntaxique d'un programme.

Durant la transformation du programme en arbre syntaxique par un parseur equipé des procédures sémantiques, les attributs hérités (p. ex. le type forcé par une conversion explicite, ou une information contextuelle, comme la position d'un item) descendent de la racine dans la direction des feuilles, tandis que les attributs synthétisés montent dans la direction de la racine. La gestion «naturelle» des attributs préconise l'usage d'un parseur récursif, descendant, ce qui permet de transmettre (par l'intermédiaire des paramètres) les attributs hérités.

Mais un parseur ascendant, p. ex. LR(1) «ne connaît pas» la racine de l'arbre, et le flot des attributs hérités devient antithétique par rapport à la propagation des valeurs gérées par le parseur. Quelques générateurs de parseurs orthodoxes interdisent l'usage des attributs hérités, mais la solution beaucoup plus universelle et élégante est possible aussi, grâce à la programmation paresseuse.

Thomas Johnsson dans l'article [14] analyse cette solution, et avoue que les sources de son inspiration sont les programmes circulaires de Bird. Analysons la règle syntaxique qui construit une expression à partir de deux sous-expressions et d'un opérateur :

$E ::= E_1 \ \mathbf{Op} \ E_2$

Cette règle pilote la synthèse des attributs de E, mais c'est également ici où les attributs hérités de E_1 et E_2 sont construits. Dénotons par E_S un attribut synthétisé attaché à l'expression E, par exemple sa «valeur», et par E_k_I des attributs hérités. Alors l'ensemble de décorations sémantiques (affectations des attributs) peut être remplacé par la création d'un attribut synthétisé E_f qui est un objet fonctionnel défini par le programme ci-dessous (en supposant que chaque variable possède deux attributs synthetisés et un hérité) :

$$
\begin{array}{lll}
E\_f = \lambda & E\_I \rightarrow & \\
 & \mathbf{let} & (E_1\_S_1, E_1\_S_2) = E_1\_f\ (E_1\_I) \\
 & & (E_2\_S_1, E_2\_S_2) = E_2\_f\ (E_2\_I) \\
 & & \{\ldots \text{ définitions des attributs } \ldots\} \\
 & \mathbf{in} & (E\_S_1, E\_S_2)
\end{array}
$$

où nous voyons que typiquement E_S dépend de E_k_S, et puisque E_k_I dépend des attributs de E, les définitions sont croisées. Johnsson utilise l'évaluation paresseuse pour résoudre de manière effective la propagation des attributs, mais il fait mieux : il exploite les attributs comme un paradigme universel, applicable en tant qu'une technique de programmation générale. Il réconstruit dans son article quelques programmes circulaires de Bird avec une simplicité et élégance remarquables. Il est vraiment amusant de voir comment la «perversion» de la monade ST suggère le même style de programmation, mais il est encore plus amusant de découvrir que les programmeurs en Fortran en ont besoin, et qu'ils utilisent des techniques analogues déjà une bonne dizaine d'années, à l'aide des trucs de programmation très pénibles.

Le lecteur intéressé par la gestion paresseuse des attributs trouvera beaucoup d'informations dans la documentation du système de compilation Elegant de Lex Augusteijn [15].

3. Construction du mode inverse de DA

3.1. Arithmétique des adjoints

Dans notre construction – comme le lecteur a déjà a pu déduire – les adjoints constituent la paramétrisation de l'état, cet état paradoxal qui se propage du futur vers le passé.

Le style monadique classique utilise souvent une syntaxe particulière : l'opérateur «*bind*» qui enchaîne les monades (c'est la transposition de notre opérateur **>=>**), ou la forme syntaxique «*do*» qui simule un style impératif de programmation. Nous voulons éviter toute syntaxe spéciale, et écrire les programmes de manière très traditionnelle et fonctionnelle, grâce à la surcharge des opérateurs standard. Ce qui restera de

l'idée monadique est le fait que l'état est caché de la surface du programme.

Il n'y aura pas de mise à jour impérative et incrémentale des adjoints dans le programme comme dans (3). Nous construirons seulement les contributions spécifiées par cette équation : $\overline{g}\frac{\partial f}{\partial e_k}$, et le résultat sera directement la somme définie dans l'équation (9).

Pour simplicité nous commençons par la discussion du cas 1-dimensionnel. L'«état final» est l'adjoint du résultat final, c'est à dire 1. l'état initial est l'adjoint de la variable indépendante. Quand le programme démarre, et la variable de différentiation est utilisée, son adjoint apparaît aussi, mais son statut existentiel est un peu fantomal, il ne sera réellement formé qu'en fin de programme, quand le transformateur final est appliqué à 1.

Dans le cas 1-dimensionnel l'état appartient au même type que toute autre expression, d'habitude c'est un nombre flottant. Le transformateur d'états, et les générateurs des constantes et de la variable de différentiation sont définis par

```
newtype Ldif a = Ld (a->(a,a))

lCnst c = Ld (\n -> (c, 0))
lDvar x = Ld (\n -> (x, n))
```

ce qui est parfaitement intuitif : l'adjoint d'une constante n'apporte rien, et l'adjoint de x engendré par l'instruction $n = x$ est $\overline{x} = \overline{n}$. La construction **`newtype`** en Haskell définit un type physiquement synonymique avec un autre, ici – avec le type fonctionnel **`a ->(a,a)`**, mais formellement différent, ce qui est assuré par la présence de la balise **`Ld`** dans le programme source, mais qui ne laisse pas de trace pendant la compilation. En Haskell il existe une autre méthode de définition littérale des synonymes, nous aurions pu écrire :

```
type Ldif a = (a->(a,a))
```

mais l'usage de **`type`** est très restreint, en particulier il est difficile de définir des opérateurs surchargés pour un tel type (les types-synonymes en Haskell ne peuvent être des instances des classes de types).

Pour des fonctions unaires et binaires quelconques, dont les dérivées (formelles) sont connues, nous définissons leur «lifting» générique dans le domaine **`Ldif`** :

```
llift f f' (Ld pp) =
  Ld (\n->let (p,pb)=pp eb
              eb=(f' p)*n in (f p,pb))

dllift op op1' op2' (Ld pp) (Ld qq) =
  Ld (\n->let (p,pb)=pp ep;  (q,qb)=qq eq
              ep=(op1' p q)*n;  eq=(op2' p q)*n
          in (op p q, pb+qb) )
```

ce qui permet immédiatemment la construction des fonctions élémentaires étendues, par exemple dans le domaine **`Ldif`** le logarithme est défini par **`log = llift log recip`**, et le cosinus par **`cos = llift cos (negate.sin)`** . Cependant, les opérateurs arithmétiques standard sont un peu optimisés, et leur définitions sont courtes (même si un peu difficiles à lire...)

```
negate (Ld pp)=Ld (\n->let (p,pb)=pp (negate n)
                       in  (negate p,pb))
(Ld pp)+(Ld qq) = Ld (\n ->
  let (p,pb)=pp n;  (q,qb)=qq n
  in  (p+q, pb+qb) )
(Ld pp)-(Ld qq) = Ld (\n ->
  let (p,pb)=pp n;  (q,qb)=qq (negate n)
  in  (p-q, pb+qb) )

(Ld pp)*(Ld qq) = Ld (\n ->
  let (p,pb)=pp (n*q);  (q,qb)=qq (p*n)
  in  (p*q, pb+qb) )
(Ld pp)/(Ld qq) = Ld (\n ->
```

```
  let (p,pb)=pp (recip q*n);  (q,qb)=qq eq
      eq=negate (p/(q*q))*n
  in  (p/q, pb+qb) )

recip (Ld pp) = Ld (\n ->
  let (p,pb)=pp eb;  w=recip p
       eb=negate (w*w)*n   in  (w,pb))

exp (Ld pp) = Ld (\n ->
  let (p,pb)=pp (w*n); w=exp p in (w,pb))

sqrt (Ld pp) = Ld (\n ->
  let (p,pb)=pp eb; w=sqrt p
      eb=(0.5/w)*n   in  (w,pb))

-- ... etc. ...
```

Notez la présence de l'addition des dérivées pour tout opérateur binaire. C'est ici que le programme accumule les adjoints et construit la somme (9).

3.2. Comment utiliser le paquetage?

Si le point de départ est une fonction numérique, par exemple la définition d'une fonction hyperbolique :

```
ch z = let e = exp z
       in (e + recip e)/2.0
```

il faut d'abord s'assurer que la fonction est reconnue par Haskell comme polymorphe, c'est-à-dire que toutes les entités : données et opérateurs sont surchargés. La définition ci-dessus malgré la présence d'une constante explicite 2.0 satisfait cette contrainte, car Haskell automatiquement surcharge les constantes numériques (les «emballe» dans un appel implicit de **`fromDouble`** ou **`fromInteger`**).

Afin de calculer la dérivée de cette fonction par rapport à son paramètre x pour, disons, $x = 0.5$ il faut

- appeler, et extraire le transformateur du type **`Ldif`** : **`res = ch (lDvar 0.5)`** ;
- appliquer le résultat à 1 : **`paireFinale = res 1`**

ce qui donne $(1.12762597, 0.521095305)$. Bien sûr, si on applique la fonction **`ch`** à **`lCnst 0.5`**, on obtient $(1.12762597, 0.0)$.

Dans la programmation pratique on peut combiner l'extraction du transformateur d'états de la structure **`Ldif`** et son application à 1 dans une fonction d'évaluation, ou peut-être tout simplement dans la fonction d'affichage du résultat final.

4. Généralisations et applications

4.1. Dérivées d'ordre supérieur

La possibilité de calculer les dérivées d'ordre supérieur est assurée «gratuitement» par le polymorphisme de Haskell. Il suffit d'appeler, par exemple : **`ch (lDvar (lDvar 0.5)`**, et en extraire la valeur désirée. Cependant en général le mode inverse n'est pas bien adapté aux dérivées d'ordre supérieur, et leur manipulation devient vite assez pénible, il suffit d'observer que le résultat d'évaluation de **`recip (lDvar (lDvar (lDvar 1.0)))`** est $(((1.0, -1.0), (-1.0, 2.0)), ((-1.0, 2.0), (2.0, -6.0)))$. Laissons au lecteur l'analyse de l'algorithme et de sa sémantique dans le cas où le type de base n'est plus numérique, mais constitue un transformateur d'états, observons seulement que la complexité de la structure résultante croît exponentiellement avec le nombre de **`lDvar`**…

Les paquetages adaptés à Fortran ou C++ d'habitude renoncent de calculer les dérivées d'ordre supérieur à 2. Si l'utilisateur a besoin de ces dérivées dans un programme fonctionnel, nous préconisons l'usage de la méthode directe [7, 8].

4.2. Cas multi-dimensionnel

Dans la section (1.2) nous avons souligné que les techniques de DA en mode inverse sont particulièrement intéressantes dans des cas M-dimensionnels (avec M variables indépendantes, dont les adjoints doivent être calculés) irréguliers, non-géométriques, où les matrices de Jacobi sont creuses. La généralisation de la méthode proposée est directe et naturelle, ce qui permet de calculer les gradients, Jacobiens, Hessiens etc., relativement facilement. Malheureusement l'efficacité de l'algorithme en souffre beaucoup. L'indice k dans l'équation (9) parcourt toutes les dimensions, et si nous voulons garder la simplicité du codage fonctionnel, les adjoints seront des vecteurs, (très creux) tandis que dans l'approche impérative les variables adjointes restaient scalaires.

Nous avons donc rédéfini le type **`Ldif`**, en remplaçant le type des adjoints par une liste, ou plutôt par un type synonymique à une liste :

```
newtype Adj a = Ad [a]
newtype Ldif a = Ld (Adj a->(a,Adj a))
```

Les adjoints des constantes sont des listes de zéros, et la k-ième variable indépendante x_k est convertie en **`Ld (\nn->(xk,Ad[0,0,...,1,0,...]))`** (le constructeur **`lDvar`** prend un paramètre supplémentaire :la position de 1 dans la liste des adjoints).

Le reste du code subit des modifications cosmétiques, par exemple la somme des adjoints demande la construction de l'opérateur **`(+)`** surchargé qui agit sur les listes additionnant les éléments, et le produit **`n*q`** où maintenant **`n`** possède plusieurs composantes, devient **`n*>q`**, où l'opérateur (non-standard) **`(*>)`** est défini par

```
Ad n *> q = Ad (map (q*) n)
```

et créé la liste de produits de **`q`** par les éléments $\mathbf{n_k}$. (En général, dans notre paquetage cet opérateur est surchargé et sert à multiplier des suites quelconques, p. ex. des listes par des éléments de base).

Le résultat final est généré par le transformateur agissant sur la liste **`Ad[1,1,...,1]`**. Bien sûr, rien n'empêche de calculer les dérivées d'ordre supérieur aussi dans le cas multi-dimensionnel.

4.3. Une optimisation légère

Notre codage des adjoints n'est pas très efficace, même si la complexité de nos algorithmes se comporte formellement de la même façon que dans d'autres implantations du mode inverse. Nous n'avons encore essayé aucune optimisation présente dans des populaires paquetages de DA, la plupart de ces optimisations étant spécifique à la programmation impérative.

L'usage de mémoire mérite une analyse approfondie. Dans un programme paresseux les expressions différées occupent la mémoire sous forme de *thunks* – fermetures composées du code compilée et de références aux objets appartenant à l'environnement de ce code. Thunks peuvent être combinées avant d'être évaluées, ils remplacent la «bande» des programmes impératifs et leur taille peut devenir très grande lors de leur application – rappelons que *toutes* les opérations différées seront alors exécutées, et qu'il n'y ait pas de réutilisation des variables adjointes ; c'est ici qu'une optimisation s'impose.

Nous proposons ici une légère modification du formalisme, ce qui permet d'alléger un peu la surcharge causé par la suspension de *toutes* les opérations. En effet, normalement les valeurs principales peuvent être calculées immédiatemment, seulement les adjoints doivent rester sous la forme fonctionnelle, différée. On peut imaginer des exceptions, si l'algorithme codé par le programme *utilise* les dérivées pour calculer d'autre chose, par exemple pour résoudre un problème d'optimisation, mais nous n'allons pas traiter ce cas ici. Introduisons donc la structure de donnée suivante et les générateurs primitifs (le cas 1-dimensionnel est présenté) :

```
data Rdif a = Rd a (a->a)
```

```
rCnst c = Rd c (\_->0)
rDvar x = Rd x id
```

L'expression **`Rd e g`** contient directement une valeur numérique **`e`**, mais son deuxième champ est une «promesse» de fournir l'adjoint quand cette expression sera utilisé dans un contexte où l'adjoint pourra être calculé. Pour récupérer la dérivée du résultat final **`Rd r g`** il faut appliquer **`g`** à 1.

L'arithmétique est une simplification du code présenté ci-dessus. Le «lifting» générique prend la forme suivante :

```
rlift f f' (Rd p pr) =
   Rd (f p) (\r->pr(r*f' p))
drlift f f1' f2' (Rd p pr) (Rd q qr) =
   Rd (f p q)
      (\r->pr(r*f1' p q)+qr(r*f2' p q))
```

Notons que l'opérateur **`f`** est appliqué tout de suite, ce qui élimine les références croisées entre l'expression et son adjoint. Ainsi nous nous sommes éloignées du modèle anti-temporal original. L'économie de mémoire introduite par cette optimisation peut être importante, même si le temps d'exécution ne doit pas subir des modifications drastiques, car le nombre d'opérations effectives reste comparable. Les opérations arithmétiques étendues deviennent :

```
negate (Rd e _) = Rd (negate e)
                     (\r->(negate r))
(Rd p pr)+(Rd q qr)=Rd (p+q) (\r->pr(r)+qr(r))
(Rd p pr)-(Rd q qr)=Rd (p-q) (\r->pr(r)+qr(negate r))

(Rd p pr)*(Rd q qr)=Rd (p*q) (\r->pr(r*q)+qr(r*p))


(Rd p pr)/(Rd q qr)=
     Rd (p/q) (\r->pr(r/q)+qr(negate r*p/(q*q)))
recip (Rd p pr)=Rd w (\r->pr(negate r*w*w))
                where w=recip p


exp (Rd p pr) = Rd w (\r -> pr(r*w)) where w=exp p
sqrt (Rd e pr) = Rd w (\r->pr(0.5*r/w))
                where w=sqrt e
-- et, tout simplement ...
log = rlift log recip
sin = rlift sin cos
cos = rlift cos (negate . sin)
```

L'économie de mémoire peut être très substantielle, mais il ne faut pas utiliser la technique inverse sans d'autres optimisations dans des cas où la chaîne des adjoints devient trop longue. Un exemple-piège typique est l'analyse de la stabilité des algorithmes de solution des équations différentielles par rapport au changement des conditions initiales. Prenons à titre d'exemple académique une équation différentielle simple, p. ex., l'oscillateur : $y''(t) + \omega^2 y = 0$, ou $y' = \omega v$: $v' = -\omega y$, qui peut être résolu par l'algorithme d'Euler, à partir de y_0 et v_0 pour $t = t_0$ arbitraire :

$$y_{n+1} = y_n + hv_n \tag{10}$$

$$v_{n+1} = v_n - hy_n \tag{11}$$

où $h = \omega\Delta t$. Nous voulons montrer que la méthode d'Euler est instable, en affichant par exemple les valeurs de $\partial y_n/\partial y_0$ pour un n assez grand. La manière la plus courte et compacte d'écrire les solutions des équations

différentielles sous forme de suites paresseuses a été proposé par nous dans [16]. Voici le code dans ce cas, où les suites y_n et v_n sont simplement représentées par des listes :

```
y = y0 : (y + h *> v)
v = v0 : (v - h *> y)
```

Les opérateurs arithmétiques sur des listes sont surchargées et combinent les éléments correspondants. Or, si on déclare p. ex. **`y0 = rDvar 1.0`**, et **`v0`** et **`h`** comme des constantes à l'aide de **`rCnst`**, le calcul de y_{1000} et l'affichage de sa valeur principale sont presque immédiates, mais on ne peut calculer la valeur de la dérivée dans un temps raisonnable. Dans les paquetages de DA on propose de lire, interpréter, et effacer la «bande» périodiquement. Encore une fois, ceci est conceptuellement très simple dans notre formalisme, et une implantation plus sérieuse du paquetage est en cours.

5. Conclusions et perspectives

L'importance des techniques fonctionnelles de programmation dans le domaine du calcul scientifique reste toujours relativement faible. La puissance de l'inférence des types, le polymorphisme et les facilités de construction des données sont reconnus, mais trop souvent l'élégance de l'approche fonctionnelle, son affinité avec la formalisation mathématique et sa compacité de codage sont reléguées au second plan, sacrifiées au nom de l'efficacité. En particulier, les techniques qui profitent de la sémantique paresseuse sont d'habitude considérées trop lentes et gourmandes en mémoire, et de plus «non-naturelles». (Les langages fonctionnels qui ont acquis une certaine reconnaissance industrielle, comme Erlang, SML ou CAML sont stricts).

Mais les ressources humaines sont coûteuses également. Dans notre opinion le fait que les techniques fonctionnelles paresseuses constituent un *outil d'algorithmisation* très puissant et économique, et qu'il est possible d'exploiter la sémantique non-stricte de manière très agressive et non-triviale, peut favoriser l'usage des langages fonctionnels par les physiciens, ingénieurs, etc.

Notre but était plutôt méthodologique, en construisant cette maquette nous voulions montrer et expliquer un style particulier de programmation fonctionnelle dans un contexte sans doute très utile. Les résultats pratiques nous semblent très prometteurs grâce à la simplicité du codage, et un peu d'exotisme présent dans le modèle de propagation des états contre le flux du temps rend le sujet assez excitant.

Références

[1] L.B. Rall, *Automatic Differentiation – Techniques and Applications*, Springer Lecture Notes in Computer Science, Vol. 120, (1981).

[2] A. Griewank, *On Automatic Differentiation*. Éd. M. Iri and K. Tanabe, *Mathematical Programming: Recent Developments and Applications*, Kluwer, (1989), pp 83–108.

[3] D. Juedes, *A taxonomy of automatic differentiation tools*. Éd. A. Griewank and G. F. Corliss, *Automatic Differentiation of Algorithms: Theory, Implementation, and Application*, SIAM, Philadelphia, Penn., (1991), pp 315–329.

[4] A. Griewank, D. Juedes H. Mitev, J. Utke, O. Vogel, A. Walther, *ADOL-C: A Package for the Automatic Differentiation of Algorithms Written in C/C++*, ACM TOMS, **22**(2) (1996), pp. 131–167, Alg. 755.

[5] Site Web de Argonne National Laboratory (USA), consacré aux techniques de la différentiation algorithmique **`www-unix.mcs.anl.gov/autodiff`**.

[6] Graham Ronald, Knuth Donald, Patashnik Oren, *Concrete Mathematics*, Addison-Wesley, Reading, MA, (1989).

[7] Jerzy Karczmarczuk, *Functional Differentiation of Computer Programs*, Journal of Higher Order and Symbolic Computing – publication en cours. Voir aussi : Proceedings, III ACM SIGPLAN International Conference on Functional Programming, Baltimore, (1998), pp. 195–203.

[8] Jerzy Karczmarczuk, *Functional Coding of Differential Forms*, I-st Scottish Workshop on Functional Programming, Stirling, Septembre 1999.

[9] R. Giering, T. Kaminski, *Recipes for Adjoint Code Construction*, ACM Trans. On Math. Software, **24(4)**, (1998), pp. 437–474.

[10] A. Griewank, *Achieving logarithmic growth of temporal and spatial complexity in reverse automatic differentiation*, Optimization Methods and Software, **1**, (1992), pp. 35–54.

[11] P. Hovland, C. H. Bischof, D. Spiegelman, M. Casella, *Efficient Derivative Codes through Automatic Differentiation and Interface Contraction: an Application in Biostatistics*, Mathematics and Computer Science Division, Argonne National Laboratory, Preprint MCS–P491–0195, (1995).

[12] P. Wadler, *The Essence of Functional programming*, 19'th Symposium on Principles of programming Languages, Santa Fe, (1992).

[13] R.S. Bird, *Using circular programs to eliminate multiple traversals of data*, Acta Informatica **21**(4), pp. 239–250, (1984).

[14] T. Johnsson, *Attribute Grammars as a Functional Programming Paradigm*, Conference on Functional programming Languages and Computer Architecture, Portland, Proceedings: Springer LNCS 274, pp. 154–173, (1987).

[15] P. Jansen, H. Munk, et L. Augusteijn, *An introduction to Elegant*, Documentation, Philips Research Laboratories, Eindhoven, Pays Bas, (1997). Site Web : **`www.research.philips.com/generalinfo/special/elegant/elegant.html`**.

[16] Jerzy Karczmarczuk, *Traitement paresseux et optimisation des suites numériques*, Actes de la conférence JFLA'2000, INRIA, pp. 17–30, (2000).